
do 11 j=3,nl
f1=d
f2=f2+twox
d=d+1.
pl(j)=(f2*pl(j-1)-f1*pl(j-2))/d

enddo 11

endif
return
END

Multidimensional Fits

If you are measuring a single variable y as a function of more than one variable
— say, a vector of variables x, then your basis functions will be functions of a vector,
X1(x), . . . , XM (x). The χ2 merit function is now

χ2 =

N∑
i=1

[
yi −

∑M
k=1 akXk(xi)
σi

]2

(15.4.24)

All of the preceding discussion goes through unchanged, with x replaced by x. In
fact, if you are willing to tolerate a bit of programming hack, you can use the above
programs without any modification: In both lfit and svdfit, the only use made
of the array elements x(i) is that each element is in turn passed to the user-supplied
routine funcs, which duly returns the values of the basis functions at that point. If
you set x(i)=i before calling lfit or svdfit, and independently provide funcs

with the true vector values of your data points (e.g., in a COMMON block), then funcs

can translate from the fictitious x(i)’s to the actual data points before doing its work.

CITED REFERENCES AND FURTHER READING:

Bevington, P.R. 1969, Data Reduction and Error Analysis for the Physical Sciences (New York:
McGraw-Hill), Chapters 8–9.

Lawson, C.L., and Hanson, R. 1974, Solving Least Squares Problems (Englewood Cliffs, NJ:
Prentice-Hall).

Forsythe, G.E., Malcolm, M.A., and Moler, C.B. 1977, Computer Methods for Mathematical
Computations (Englewood Cliffs, NJ: Prentice-Hall), Chapter 9.

15.5 Nonlinear Models

We now consider fitting when the model depends nonlinearly on the set of M
unknown parameters ak, k = 1, 2, . . . ,M . We use the same approach as in previous
sections, namely to define a χ2 merit function and determine best-fit parameters
by its minimization. With nonlinear dependences, however, the minimization must
proceed iteratively. Given trial values for the parameters, we develop a procedure
that improves the trial solution. The procedure is then repeated until χ2 stops (or
effectively stops) decreasing.

How is this problem different from the general nonlinear function minimization
problem already dealt with in Chapter 10? Superficially, not at all: Sufficiently
close to the minimum, we expect the χ2 function to be well approximated by a
quadratic form, which we can write as

χ2(a) ≈ γ − d · a +
1

2
a · D · a (15.5.1)

where d is an M -vector and D is an M ×M matrix. (Compare equation 10.6.1.)
If the approximation is a good one, we know how to jump from the current trial
parameters acur to the minimizing ones amin in a single leap, namely

amin = acur + D−1 ·
[
−∇χ2(acur)

]
(15.5.2)

(Compare equation 10.7.4.)
On the other hand, (15.5.1) might be a poor local approximation to the shape

of the function that we are trying to minimize at acur. In that case, about all we
can do is take a step down the gradient, as in the steepest descent method (§10.6).
In other words,

anext = acur − constant×∇χ2(acur) (15.5.3)

where the constant is small enough not to exhaust the downhill direction.
To use (15.5.2) or (15.5.3), we must be able to compute the gradient of the χ2

function at any set of parameters a. To use (15.5.2) we also need the matrix D, which
is the second derivative matrix (Hessian matrix) of the χ2 merit function, at any a.

Now, this is the crucial difference from Chapter 10: There, we had no way of
directly evaluating the Hessian matrix. We were given only the ability to evaluate
the function to be minimized and (in some cases) its gradient. Therefore, we had
to resort to iterative methods not just because our function was nonlinear, but also
in order to build up information about the Hessian matrix. Sections 10.7 and 10.6
concerned themselves with two different techniques for building up this information.

Here, life is much simpler. We know exactly the form of χ2, since it is based
on a model function that we ourselves have specified. Therefore the Hessian matrix
is known to us. Thus we are free to use (15.5.2) whenever we care to do so. The
only reason to use (15.5.3) will be failure of (15.5.2) to improve the fit, signaling
failure of (15.5.1) as a good local approximation.

Calculation of the Gradient and Hessian

The model to be fitted is

y = y(x; a) (15.5.4)

and the χ2 merit function is

χ2(a) =

N∑
i=1

[
yi − y(xi; a)

σi

]2

(15.5.5)

The gradient of χ2 with respect to the parameters a, which will be zero at the χ2

minimum, has components

∂χ2

∂ak
= −2

N∑
i=1

[yi − y(xi; a)]

σ2
i

∂y(xi; a)

∂ak
k = 1, 2, . . . ,M (15.5.6)

Taking an additional partial derivative gives

∂2χ2

∂ak∂al
= 2

N∑
i=1

1

σ2
i

[
∂y(xi; a)

∂ak

∂y(xi; a)

∂al
− [yi − y(xi; a)]

∂2y(xi; a)

∂al∂ak

]
(15.5.7)

It is conventional to remove the factors of 2 by defining

βk ≡ −
1

2

∂χ2

∂ak
αkl ≡

1

2

∂2χ2

∂ak∂al
(15.5.8)

making [α] = 1
2 D in equation (15.5.2), in terms of which that equation can be

rewritten as the set of linear equations

M∑
l=1

αkl δal = βk (15.5.9)

This set is solved for the increments δal that, added to the current approximation,
give the next approximation. In the context of least-squares, the matrix [α], equal to
one-half times the Hessian matrix, is usually called the curvature matrix.

Equation (15.5.3), the steepest descent formula, translates to

δal = constant× βl (15.5.10)

Note that the componentsαkl of the Hessian matrix (15.5.7) depend both on the
first derivatives and on the second derivatives of the basis functions with respect to
their parameters. Some treatments proceed to ignore the second derivative without
comment. We will ignore it also, but only after a few comments.

Second derivatives occur because the gradient (15.5.6) already has a dependence
on ∂y/∂ak , so the next derivative simply must contain terms involving∂2y/∂al∂ak.
The second derivative term can be dismissed when it is zero (as in the linear case
of equation 15.4.8), or small enough to be negligible when compared to the term
involving the first derivative. It also has an additional possibility of being ignorably
small in practice: The term multiplying the second derivative in equation (15.5.7)
is [yi − y(xi; a)]. For a successful model, this term should just be the random
measurement error of each point. This error can have either sign, and should in
general be uncorrelated with the model. Therefore, the second derivative terms tend
to cancel out when summed over i.

Inclusion of the second-derivative term can in fact be destabilizing if the model
fits badly or is contaminated by outlier points that are unlikely to be offset by

compensating points of opposite sign. From this point on, we will always use as
the definition of αkl the formula

αkl =

N∑
i=1

1

σ2
i

[
∂y(xi; a)

∂ak

∂y(xi; a)

∂al

]
(15.5.11)

This expression more closely resembles its linear cousin (15.4.8). You should
understand that minor (or even major) fiddling with [α] has no effect at all on
what final set of parameters a is reached, but affects only the iterative route that is
taken in getting there. The condition at the χ2 minimum, that βk = 0 for all k,
is independent of how [α] is defined.

Levenberg-Marquardt Method

Marquardt [1] has put forth an elegant method, related to an earlier suggestion
of Levenberg, for varying smoothly between the extremes of the inverse-Hessian
method (15.5.9) and the steepest descent method (15.5.10). The latter method is
used far from the minimum, switching continuously to the former as the minimum
is approached. This Levenberg-Marquardt method (also called Marquardt method)
works very well in practice and has become the standard of nonlinear least-squares
routines.

The method is based on two elementary, but important, insights. Consider the
“constant” in equation (15.5.10). What should it be, even in order of magnitude?
What sets its scale? There is no information about the answer in the gradient. That
tells only the slope, not how far that slope extends. Marquardt’s first insight is that
the components of the Hessian matrix, even if they are not usable in any precise
fashion, give some information about the order-of-magnitude scale of the problem.

The quantity χ2 is nondimensional, i.e., is a pure number; this is evident from
its definition (15.5.5). On the other hand, βk has the dimensions of 1/ak, which
may well be dimensional, i.e., have units like cm−1, or kilowatt-hours, or whatever.
(In fact, each component of βk can have different dimensions!) The constant of
proportionality between βk and δak must therefore have the dimensions of a2

k . Scan
the components of [α] and you see that there is only one obvious quantity with these
dimensions, and that is 1/αkk, the reciprocal of the diagonal element. So that must
set the scale of the constant. But that scale might itself be too big. So let’s divide
the constant by some (nondimensional) fudge factor λ, with the possibility of setting
λ � 1 to cut down the step. In other words, replace equation (15.5.10) by

δal =
1

λαll
βl or λαll δal = βl (15.5.12)

It is necessary that all be positive, but this is guaranteed by definition (15.5.11) —
another reason for adopting that equation.

Marquardt’s second insight is that equations (15.5.12) and (15.5.9) can be
combined if we define a new matrix α′ by the following prescription

α′jj ≡ αjj(1 + λ)

α′jk ≡ αjk (j 6= k)
(15.5.13)

and then replace both (15.5.12) and (15.5.9) by

M∑
l=1

α′kl δal = βk (15.5.14)

When λ is very large, the matrix α′ is forced into being diagonally dominant, so
equation (15.5.14) goes over to be identical to (15.5.12). On the other hand, as λ
approaches zero, equation (15.5.14) goes over to (15.5.9).

Given an initial guess for the set of fitted parameters a, the recommended
Marquardt recipe is as follows:

• Compute χ2(a).
• Pick a modest value for λ, say λ = 0.001.
• (†) Solve the linear equations (15.5.14) for δa and evaluate χ2(a + δa).
• If χ2(a + δa) ≥χ2(a), increase λ by a factor of 10 (or any other

substantial factor) and go back to (†).
• If χ2(a + δa) < χ2(a), decrease λ by a factor of 10, update the trial

solution a ← a + δa, and go back to (†).
Also necessary is a condition for stopping. Iterating to convergence (to machine

accuracy or to the roundoff limit) is generally wasteful and unnecessary since the
minimum is at best only a statistical estimate of the parameters a. As we will see
in §15.6, a change in the parameters that changes χ2 by an amount � 1 is never
statistically meaningful.

Furthermore, it is not uncommon to find the parameters wandering
around near the minimum in a flat valley of complicated topography. The rea-
son is that Marquardt’s method generalizes the method of normal equations (§15.4),
hence has the same problem as that method with regard to near-degeneracy of the
minimum. Outright failure by a zero pivot is possible, but unlikely. More often,
a small pivot will generate a large correction which is then rejected, the value of
λ being then increased. For sufficiently large λ the matrix [α′] is positive definite
and can have no small pivots. Thus the method does tend to stay away from zero
pivots, but at the cost of a tendency to wander around doing steepest descent in
very un-steep degenerate valleys.

These considerations suggest that, in practice, one might as well stop iterating
on the first or second occasion that χ2 decreases by a negligible amount, say either
less than 0.01 absolutely or (in case roundoff prevents that being reached) some
fractional amount like 10−3. Don’t stop after a step where χ2 increases: That only
shows that λ has not yet adjusted itself optimally.

Once the acceptable minimum has been found, one wants to set λ = 0 and
compute the matrix

[C] ≡ [α]−1 (15.5.15)

which, as before, is the estimated covariance matrix of the standard errors in the
fitted parameters a (see next section).

The following pair of subroutines encodes Marquardt’s method for nonlinear
parameter estimation. Much of the organization matches that used in lfit of
§15.4. In particular the array ia(1:ma) must be input with components one or zero
corresponding to whether the respective parameter values a(1:ma) are to be fitted
for or held fixed at their input values, respectively.

The routine mrqmin performs one iteration of Marquardt’s method. It is first
called (once) with alamda < 0, which signals the routine to initialize. alamda is
returned on the first and all subsequent calls as the suggested value of λ for the
next iteration; a and chisq are always returned as the best parameters found so
far and their χ2. When convergence is deemed satisfactory, set alamda to zero
before a final call. The matrices alpha and covar (which were used as workspace
in all previous calls) will then be set to the curvature and covariance matrices for
the converged parameter values. The arguments alpha, a, and chisq must not be
modified between calls, nor should alamda be, except to set it to zero for the final
call. When an uphill step is taken, chisq and a are returned with their input (best)
values, but alamda is returned with an increased value.

The routine mrqmin calls the routine mrqcof for the computation of the matrix
[α] (equation 15.5.11) and vector β (equations 15.5.6 and 15.5.8). In turn mrqcof

calls the user-supplied routine funcs(x,a,y,dyda), which for input values x ≡ xi
and a ≡ a returns the model function y ≡ y(xi; a) and the vector of derivatives
dyda ≡ ∂y/∂ak .

SUBROUTINE mrqmin(x,y,sig,ndata,a,ia,ma,covar,alpha,nca,
* chisq,funcs,alamda)

INTEGER ma,nca,ndata,ia(ma),MMAX
REAL alamda,chisq,funcs,a(ma),alpha(nca,nca),covar(nca,nca),

* sig(ndata),x(ndata),y(ndata)
PARAMETER (MMAX=20) Set to largest number of fit parameters.

C USES covsrt,gaussj,mrqcof
Levenberg-Marquardt method, attempting to reduce the value χ2 of a fit between a set of
data points x(1:ndata), y(1:ndata)with individual standard deviations sig(1:ndata),
and a nonlinear function dependent on ma coefficients a(1:ma). The input array ia(1:ma)
indicates by nonzero entries those components of a that should be fitted for, and by zero
entries those components that should be held fixed at their input values. The program
returns current best-fit values for the parameters a(1:ma), and χ2 = chisq. The ar-
rays covar(1:nca,1:nca), alpha(1:nca,1:nca) with physical dimension nca (≥ the
number of fitted parameters) are used as working space during most iterations. Supply a
subroutine funcs(x,a,yfit,dyda,ma) that evaluates the fitting function yfit, and its
derivatives dyda with respect to the fitting parameters a at x. On the first call provide
an initial guess for the parameters a, and set alamda<0 for initialization (which then sets
alamda=.001). If a step succeeds chisq becomes smaller and alamda decreases by a
factor of 10. If a step fails alamda grows by a factor of 10. You must call this routine
repeatedly until convergence is achieved. Then, make one final call with alamda=0, so
that covar(1:ma,1:ma) returns the covariance matrix, and alpha the curvature matrix.
(Parameters held fixed will return zero covariances.)

INTEGER j,k,l,mfit
REAL ochisq,atry(MMAX),beta(MMAX),da(MMAX)
SAVE ochisq,atry,beta,da,mfit
if(alamda.lt.0.)then Initialization.

mfit=0
do 11 j=1,ma

if (ia(j).ne.0) mfit=mfit+1
enddo 11

alamda=0.001
call mrqcof(x,y,sig,ndata,a,ia,ma,alpha,beta,nca,chisq,funcs)
ochisq=chisq
do 12 j=1,ma

atry(j)=a(j)
enddo 12

endif
do 14 j=1,mfit Alter linearized fitting matrix, by augmenting

diagonal elements.do 13 k=1,mfit

covar(j,k)=alpha(j,k)
enddo 13

covar(j,j)=alpha(j,j)*(1.+alamda)
da(j)=beta(j)

enddo 14

call gaussj(covar,mfit,nca,da,1,1) Matrix solution.
if(alamda.eq.0.)then Once converged, evaluate covariance matrix.

call covsrt(covar,nca,ma,ia,mfit)
call covsrt(alpha,nca,ma,ia,mfit) Spread out alpha to its full size too.
return

endif
j=0
do 15 l=1,ma Did the trial succeed?

if(ia(l).ne.0) then
j=j+1
atry(l)=a(l)+da(j)

endif
enddo 15

call mrqcof(x,y,sig,ndata,atry,ia,ma,covar,da,nca,chisq,funcs)
if(chisq.lt.ochisq)then Success, accept the new solution.

alamda=0.1*alamda
ochisq=chisq
do 17 j=1,mfit

do 16 k=1,mfit
alpha(j,k)=covar(j,k)

enddo 16

beta(j)=da(j)
enddo 17

do 18 l=1,ma
a(l)=atry(l)

enddo 18

else Failure, increase alamda and return.
alamda=10.*alamda
chisq=ochisq

endif
return
END

Notice the use of the routine covsrt from §15.4. This is merely for rearranging
the covariance matrix covar into the order of all ma parameters. The above routine
also makes use of

SUBROUTINE mrqcof(x,y,sig,ndata,a,ia,ma,alpha,beta,nalp,
* chisq,funcs)

INTEGER ma,nalp,ndata,ia(ma),MMAX
REAL chisq,a(ma),alpha(nalp,nalp),beta(ma),sig(ndata),x(ndata),

* y(ndata)
EXTERNAL funcs
PARAMETER (MMAX=20)

Used by mrqmin to evaluate the linearized fitting matrix alpha, and vector beta as in
(15.5.8), and calculate χ2.

INTEGER mfit,i,j,k,l,m
REAL dy,sig2i,wt,ymod,dyda(MMAX)
mfit=0
do 11 j=1,ma

if (ia(j).ne.0) mfit=mfit+1
enddo 11

do 13 j=1,mfit Initialize (symmetric) alpha, beta.
do 12 k=1,j

alpha(j,k)=0.
enddo 12

beta(j)=0.
enddo 13

chisq=0.
do 16 i=1,ndata Summation loop over all data.

call funcs(x(i),a,ymod,dyda,ma)
sig2i=1./(sig(i)*sig(i))
dy=y(i)-ymod
j=0
do 15 l=1,ma

if(ia(l).ne.0) then
j=j+1
wt=dyda(l)*sig2i
k=0
do 14 m=1,l

if(ia(m).ne.0) then
k=k+1
alpha(j,k)=alpha(j,k)+wt*dyda(m)

endif
enddo 14

beta(j)=beta(j)+dy*wt
endif

enddo 15

chisq=chisq+dy*dy*sig2i And find χ2.
enddo 16

do 18 j=2,mfit Fill in the symmetric side.
do 17 k=1,j-1

alpha(k,j)=alpha(j,k)
enddo 17

enddo 18

return
END

Example

The following subroutine fgauss is an example of a user-supplied subroutine
funcs. Used with the above routine mrqmin (in turn using mrqcof, covsrt, and
gaussj), it fits for the model

y(x) =

K∑
k=1

Bk exp

[
−
(
x−Ek
Gk

)2
]

(15.5.16)

which is a sum of K Gaussians, each having a variable position, amplitude, and
width. We store the parameters in the order B1, E1, G1, B2, E2, G2, . . . , BK ,
EK , GK .

SUBROUTINE fgauss(x,a,y,dyda,na)
INTEGER na
REAL x,y,a(na),dyda(na)

y(x; a) is the sum of na/3 Gaussians (15.5.16). The amplitude, center, and width of the
Gaussians are stored in consecutive locations of a: a(i) = Bk, a(i+1) = Ek, a(i+2) =
Gk , k = 1, ..., na/3.

INTEGER i
REAL arg,ex,fac
y=0.
do 11 i=1,na-1,3

arg=(x-a(i+1))/a(i+2)
ex=exp(-arg**2)
fac=a(i)*ex*2.*arg
y=y+a(i)*ex
dyda(i)=ex
dyda(i+1)=fac/a(i+2)
dyda(i+2)=fac*arg/a(i+2)

enddo 11

return
END

More Advanced Methods for Nonlinear Least Squares

The Levenberg-Marquardt algorithm can be implemented as a model-trust
region method for minimization (see §9.7 and ref. [2]) applied to the special case
of a least squares function. A code of this kind due to Moré [3] can be found in
MINPACK [4]. Another algorithm for nonlinear least-squares keeps the second-
derivative term we dropped in the Levenberg-Marquardt method whenever it would
be better to do so. These methods are called “full Newton-type” methods and are
reputed to be more robust than Levenberg-Marquardt, but more complex. One
implementation is the code NL2SOL [5].

15.6 Confidence Limits on Estimated Model
Parameters

Several times already in this chapter we have made statements about the standard
errors, or uncertainties, in a set of M estimated parameters a. We have given some
formulas for computing standard deviations or variances of individual parameters
(equations 15.2.9, 15.4.15, 15.4.19), as well as some formulas for covariances
between pairs of parameters (equation 15.2.10; remark following equation 15.4.15;
equation 15.4.20; equation 15.5.15).

In this section, we want to be more explicit regarding the precise meaning
of these quantitative uncertainties, and to give further information about how
quantitative confidence limits on fitted parameters can be estimated. The subject
can get somewhat technical, and even somewhat confusing, so we will try to make
precise statements, even when they must be offered without proof.

Figure 15.6.1 shows the conceptual scheme of an experiment that “measures”
a set of parameters. There is some underlying true set of parameters atrue that are
known to Mother Nature but hidden from the experimenter. These true parameters
are statistically realized, along with random measurement errors, as a measured data
set, which we will symbolize asD(0). The data setD(0) is known to the experimenter.
He or she fits the data to a model by χ2 minimization or some other technique, and
obtains measured, i.e., fitted, values for the parameters, which we here denote a(0).

Because measurement errors have a random component, D(0) is not a unique
realization of the true parameters atrue. Rather, there are infinitely many other
realizations of the true parameters as “hypothetical data sets” each of which could
have been the one measured, but happened not to be. Let us symbolize these
by D(1),D(2), . . . . Each one, had it been realized, would have given a slightly
different set of fitted parameters, a(1), a(2), . . . , respectively. These parameter sets
a(i) therefore occur with some probability distribution in the M -dimensional space
of all possible parameter sets a. The actual measured set a(0) is one member drawn
from this distribution.

Even more interesting than the probability distribution of a(i) would be the
distribution of the difference a(i) − atrue. This distribution differs from the former
one by a translation that puts Mother Nature’s true value at the origin. If we knew this
distribution, we would know everything that there is to know about the quantitative
uncertainties in our experimental measurement a(0).

So the name of the game is to find some way of estimating or approximating
the probability distribution of a(i)− atrue without knowing atrue and without having
available to us an infinite universe of hypothetical data sets.

Monte Carlo Simulation of Synthetic Data Sets

Although the measured parameter set a(0) is not the true one, let us consider
a fictitious world in which it was the true one. Since we hope that our measured
parameters are not too wrong, we hope that that fictitious world is not too different
from the actual world with parameters atrue. In particular, let us hope — no, let us
assume — that the shape of the probability distribution a(i) − a(0) in the fictitious
world is the same, or very nearly the same, as the shape of the probability distribution


